

# NAG Library Routine Document

## G02BLF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of **bold italicised** terms and other implementation-dependent details.

### 1 Purpose

G02BLF computes means and standard deviations, sums of squares and cross-products about zero, and correlation-like coefficients for selected variables omitting completely any cases with a missing observation for any variable (either over all variables in the dataset or over only those variables in the selected subset).

### 2 Specification

```

SUBROUTINE G02BLF (N, M, X, LDX, MISS, XMISS, MISTYP, NVAR, KVAR, XBAR,      &
                  STD, SSPZ, LDSSPZ, RZ, LDRZ, NCASES, IFAIL)
INTEGER          N, M, LDX, MISS(M), MISTYP, NVAR, KVAR(NVAR), LDSSPZ,    &
                LDRZ, NCASES, IFAIL
REAL (KIND=nag_wp) X(LDX,M), XMISS(M), XBAR(NVAR), STD(NVAR),          &
                SSPZ(LDSSPZ,NVAR), RZ(LDRZ,NVAR)

```

### 3 Description

The input data consist of  $n$  observations for each of  $m$  variables, given as an array

$$[x_{ij}], \quad i = 1, 2, \dots, n (n \geq 2) \text{ and } j = 1, 2, \dots, m (m \geq 2),$$

where  $x_{ij}$  is the  $i$ th observation on the  $j$ th variable, together with the subset of these variables,  $v_1, v_2, \dots, v_p$ , for which information is required.

In addition, each of the  $m$  variables may optionally have associated with it a value which is to be considered as representing a missing observation for that variable; the missing value for the  $j$ th variable is denoted by  $xm_j$ . Missing values need not be specified for all variables.

The missing values can be utilized in two slightly different ways, you can indicate which scheme is required.

Firstly, let  $w_i = 0$  if observation  $i$  contains a missing value for any of those variables in the set  $1, 2, \dots, m$  for which missing values have been declared, i.e., if  $x_{ij} = xm_j$  for any  $j$  ( $j = 1, 2, \dots, m$ ) for which an  $xm_j$  has been assigned (see also Section 7); and  $w_i = 1$  otherwise, for  $i = 1, 2, \dots, n$ .

Secondly, let  $w_i = 0$  if observation  $i$  contains a missing value for any of those variables in the selected subset  $v_1, v_2, \dots, v_p$  for which missing values have been declared, i.e., if  $x_{ij} = xm_j$  for any  $j$  ( $j = v_1, v_2, \dots, v_p$ ) for which an  $xm_j$  has been assigned (see also Section 7); and  $w_i = 1$  otherwise, for  $i = 1, 2, \dots, n$ .

The quantities calculated are:

(a) Means:

$$\bar{x}_j = \frac{\sum_{i=1}^n w_i x_{ij}}{\sum_{i=1}^n w_i}, \quad j = v_1, v_2, \dots, v_p.$$

(b) Standard deviations:

$$S_j = \sqrt{\frac{\sum_{i=1}^n w_i (x_{ij} - \bar{x}_j)^2}{\sum_{i=1}^n w_i - 1}}, \quad j = v_1, v_2, \dots, v_p.$$

(c) Sums of squares and cross-products about zero:

$$\tilde{S}_{jk} = \sum_{i=1}^n w_i x_{ij} x_{ik}, \quad j, k = v_1, v_2, \dots, v_p.$$

(d) Correlation-like coefficients:

$$\tilde{R}_{jk} = \frac{\tilde{S}_{jk}}{\sqrt{\tilde{S}_{jj} \tilde{S}_{kk}}}, \quad j, k = v_1, v_2, \dots, v_p.$$

If  $\tilde{S}_{jj}$  or  $\tilde{S}_{kk}$  is zero,  $\tilde{R}_{jk}$  is set to zero.

## 4 References

None.

## 5 Parameters

- 1: N – INTEGER *Input*  
*On entry:*  $n$ , the number of observations or cases.  
*Constraint:*  $N \geq 2$ .
- 2: M – INTEGER *Input*  
*On entry:*  $m$ , the number of variables.  
*Constraint:*  $M \geq 2$ .
- 3: X(LDX,M) – REAL (KIND=nag\_wp) array *Input*  
*On entry:*  $X(i, j)$  must be set to  $x_{ij}$ , the value of the  $i$ th observation on the  $j$ th variable, for  $i = 1, 2, \dots, n$  and  $j = 1, 2, \dots, m$ .
- 4: LDX – INTEGER *Input*  
*On entry:* the first dimension of the array X as declared in the (sub)program from which G02BLF is called.  
*Constraint:*  $LDX \geq N$ .
- 5: MISS(M) – INTEGER array *Input/Output*  
*On entry:*  $MISS(j)$  must be set equal to 1 if a missing value,  $xm_j$ , is to be specified for the  $j$ th variable in the array X, or set equal to 0 otherwise. Values of MISS must be given for all  $m$  variables in the array X.  
*On exit:* the array MISS is overwritten by the routine, and the information it contained on entry is lost.

- 6: XMISS(M) – REAL (KIND=nag\_wp) array *Input/Output*  
*On entry:* XMISS( $j$ ) must be set to the missing value,  $xm_j$ , to be associated with the  $j$ th variable in the array X, for those variables for which missing values are specified by means of the array MISS (see Section 7).  
*On exit:* the array XMISS is overwritten by the routine, and the information it contained on entry is lost.
- 7: MISTYP – INTEGER *Input*  
*On entry:* indicates the manner in which missing observations are to be treated.  
MISTYP = 1  
A case is excluded if it contains a missing value for any of the variables  $1, 2, \dots, m$ .  
MISTYP = 0  
A case is excluded if it contains a missing value for any of the  $p(\leq m)$  variables specified in the array KVAR.
- 8: NVAR – INTEGER *Input*  
*On entry:*  $p$ , the number of variables for which information is required.  
*Constraint:*  $2 \leq \text{NVAR} \leq M$ .
- 9: KVAR(NVAR) – INTEGER array *Input*  
*On entry:* KVAR( $j$ ) must be set to the column number in X of the  $j$ th variable for which information is required, for  $j = 1, 2, \dots, p$ .  
*Constraint:*  $1 \leq \text{KVAR}(j) \leq M$ , for  $j = 1, 2, \dots, p$ .
- 10: XBAR(NVAR) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* the mean value,  $\bar{x}_j$ , of the variable specified in KVAR( $j$ ), for  $j = 1, 2, \dots, p$ .
- 11: STD(NVAR) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* the standard deviation,  $s_j$ , of the variable specified in KVAR( $j$ ), for  $j = 1, 2, \dots, p$ .
- 12: SSPZ(LDSSPZ, NVAR) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* SSPZ( $j, k$ ) is the cross-product about zero,  $\tilde{S}_{jk}$ , for the variables specified in KVAR( $j$ ) and KVAR( $k$ ), for  $j = 1, 2, \dots, p$  and  $k = 1, 2, \dots, p$ .
- 13: LDSSPZ – INTEGER *Input*  
*On entry:* the first dimension of the array SSPZ as declared in the (sub)program from which G02BLF is called.  
*Constraint:* LDSSPZ  $\geq$  NVAR.
- 14: RZ(LDRZ, NVAR) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* RZ( $j, k$ ) is the correlation-like coefficient,  $\tilde{R}_{jk}$ , between the variables specified in KVAR( $j$ ) and KVAR( $k$ ), for  $j = 1, 2, \dots, p$  and  $k = 1, 2, \dots, p$ .
- 15: LDRZ – INTEGER *Input*  
*On entry:* the first dimension of the array RZ as declared in the (sub)program from which G02BLF is called.  
*Constraint:* LDRZ  $\geq$  NVAR.

16: NCASES – INTEGER *Output*

*On exit:* the number of cases actually used in the calculations (when cases involving missing values have been eliminated).

17: IFAIL – INTEGER *Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $N < 2$ .

IFAIL = 2

On entry,  $NVARS < 2$ ,  
or  $NVARS > M$ .

IFAIL = 3

On entry,  $LDX < N$ ,  
or  $LDSSPZ < NVARS$ ,  
or  $LDRZ < NVARS$ .

IFAIL = 4

On entry,  $KVAR(j) < 1$ ,  
or  $KVAR(j) > M$  for some  $j = 1, 2, \dots, NVARS$ .

IFAIL = 5

On entry,  $MISTYP \neq 1$  or 0.

IFAIL = 6

After observations with missing values were omitted, no cases remained.

IFAIL = 7

After observations with missing values were omitted, only one case remained.

## 7 Accuracy

G02BLF does not use *additional precision* arithmetic for the accumulation of scalar products, so there may be a loss of significant figures for large  $n$ .

You are warned of the need to exercise extreme care in your selection of missing values. G02BLF treats all values in the inclusive range  $\left(1 \pm 0.1^{(X02BEF-2)}\right) \times xm_j$ , where  $xm_j$  is the missing value for variable  $j$  specified in XMISS.

You must therefore ensure that the missing value chosen for each variable is sufficiently different from all valid values for that variable so that none of the valid values fall within the range indicated above.

## 8 Further Comments

The time taken by G02BLF depends on  $n$  and  $p$ , and the occurrence of missing values.

The routine uses a two-pass algorithm.

## 9 Example

This example reads in a set of data consisting of five observations on each of four variables. Missing values of 0.0 are declared for the second and fourth variables; no missing values are specified for the first and third variables. The means, standard deviations, sums of squares and cross-products about zero, and correlation-like coefficients for the fourth, first and second variables are then calculated and printed, omitting completely all cases containing missing values for these three selected variables; cases 3 and 4 are therefore eliminated, leaving only three cases in the calculations.

### 9.1 Program Text

```

Program g02blfe

!      G02BLF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
Use nag_library, Only: g02blf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Integer                     :: i, ifail, ldrz, ldsspz, ldx, m,      &
                             mistyp, n, ncases, nvars
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: rz(:, :), sspz(:, :), std(:), x(:, :), &
                             xbar(:), xmiss(:)
Integer, Allocatable          :: kvar(:), miss(:)
!      .. Executable Statements ..
Write (nout,*) 'G02BLF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Read in the problem size
Read (nin,*) n, m, nvars, mistyp

      ldrz = nvars
      ldsspz = nvars
      ldx = n
      Allocate (rz(ldrz,nvars),sspz(ldsspz,nvars),std(nvars),x(ldx,m), &
               xbar(nvars),xmiss(m),kvar(nvars),miss(m))

!      Read in data
Read (nin,*)(x(i,1:m),i=1,n)

!      Read in missing value flags
Read (nin,*) miss(1:m)
Read (nin,*) xmiss(1:m)

```

```

!      Read in column IDs
      Read (nin,*) kvar(1:nvars)

!      Display data
      Write (nout,99999) 'Number of variables (columns) =', m
      Write (nout,99999) 'Number of cases      (rows)      =', n
      Write (nout,*)
      Write (nout,*) 'Data matrix is:-'
      Write (nout,*)
      Write (nout,99998)(i,i=1,m)
      Write (nout,99997)(i,x(i,1:m),i=1,n)
      Write (nout,*)

!      Calculate summary statistics
      ifail = 0
      Call g02blf(n,m,x,ldx,miss,xmiss,mistyp,nvars,kvar,xbar,std,sspz,ldsspz, &
        rz,ldrz,ncases,ifail)

!      Display results
      Write (nout,*) 'Variable   Mean     St. dev.'
      Write (nout,99995)(kvar(i),xbar(i),std(i),i=1,nvars)
      Write (nout,*)
      Write (nout,*) 'Sums of squares and cross-products about' // ' zero'
      Write (nout,99998) kvar(1:nvars)
      Write (nout,99996)(kvar(i),sspz(i,1:nvars),i=1,nvars)
      Write (nout,*)
      Write (nout,*) 'Correlation-like coefficients'
      Write (nout,99998) kvar(1:nvars)
      Write (nout,99996)(kvar(i),rz(i,1:nvars),i=1,nvars)
      Write (nout,*)
      Write (nout,99999) 'Number of cases actually used:', ncases

99999 Format (1X,A,I5)
99998 Format (1X,4I12)
99997 Format (1X,I3,4F12.4)
99996 Format (1X,I3,3F12.4)
99995 Format (1X,I5,2F11.4)
      End Program g02blfe

```

## 9.2 Program Data

```

G02BLF Example Program Data
5 4 3 0      :: N, M, NVARs, MISTYP
 3.0  3.0  1.0  2.0
 6.0  4.0 -1.0  4.0
 9.0  0.0  5.0  9.0
12.0  2.0  0.0  0.0
-1.0  5.0  4.0 12.0  :: End of X
 0     1     0     1  :: MISS
 0.0  0.0  0.0  0.0  :: XMISS
4 1 2      :: KVAR

```

## 9.3 Program Results

G02BLF Example Program Results

```

Number of variables (columns) = 4
Number of cases      (rows)   = 5

```

Data matrix is:-

	1	2	3	4
1	3.0000	3.0000	1.0000	2.0000
2	6.0000	4.0000	-1.0000	4.0000
3	9.0000	0.0000	5.0000	9.0000
4	12.0000	2.0000	0.0000	0.0000
5	-1.0000	5.0000	4.0000	12.0000

```

Variable   Mean     St. dev.
 4         6.0000  5.2915

```

1	2.6667	3.5119
2	4.0000	1.0000

Sums of squares and cross-products about zero

	4	1	2
4	164.0000	18.0000	82.0000
1	18.0000	46.0000	28.0000
2	82.0000	28.0000	50.0000

Correlation-like coefficients

	4	1	2
4	1.0000	0.2072	0.9055
1	0.2072	1.0000	0.5838
2	0.9055	0.5838	1.0000

Number of cases actually used: 3